IDENTIFICATION

Product Code:	MAINDEC-08-D2GA-D
Product Name:	PDP-8 High Speed Punch Test
Date Created:	May 12, 1965
Maintainer:	Diagnostic Group
Author:	M. Horovitz
Previous Code:	MAINDEC 817

1. ABSTRACT

This program consists of two separate tests. The first causes the High Speed Punch to produce a tape containing a sequence of "pseudo-random" character codes. This tape is checked for accuracy using either the high-speed reader or the Teletype reader.

In the second test, the character code represented by the setting of SR_{4-11} is punched repeatedly. The switch setting may be changed while the test is running.

The interrupt system is also tested.

2. REQUIREMENTS

Storage

The test program occupies memory locations 0001-0163 and requires that the RIM loader be in locations 7756-7776.

Equipment

Standard PDP-8 with Teletype Keyboard/Printer

High-Speed Tape Punch and Control

(Optional) High-Speed Tape Reader and Control Type 750

3. USAGE

3.1 Loading

The RIM Loader must be in memory.

Turn off the Teletype reader

Set the SWITCH REGISTER to 7756; press LOAD ADDRESS, then START.

Place the RIM program tape in the Teletype reader, and turn on the reader.

When the program has been loaded, stop the computer, turn off the reader, and remove the tape.

3.2 Switch Settings

SW_0 Down (off). Causes program to punch the test tape.

SW_0 Up (on). Terminates punching and conditions the program to begin checking the punched tape.

3.3 Start Up and/or Entry

With the program in memory, set the SWITCH REGISTER to the desired starting address as follows:

0020 Entry when using the high-speed reader for checking.

0021 Entry when using the Teletype reader for checking.

0022 Starting address of the Switch Register Test.

Press LOAD ADDRESS.

Put SW_0 down.

(For SR punch test) Place the desired character code in SR_{4-11}.

Press START.

As long as SW_0 is down, the punch will continue to produce a pseudo-random sequence. As soon as the switch is raised, the punching stops, a length of blank trailer is produced, and the program halts. At this point, the C(MA) will be 0041. Remove the punched tape from the bin and place it in the reader chosen to do the checking. The blank leader must be positioned at the read heads so that the checking program will not miss the first character punched.

If using the high-speed reader, turn on reader power, place the guide arm down, and press CONTINUE. The Teletype reader must be off.

When using the Teletype reader, with the tape in place press CONTINUE, then turn on the reader.

The checking program will continue until an error occurs or until the blank trailer is encountered.

3.4 Errors

Stop	C(MA)	Cause of Stop
E1	0011	The PLS instruction did not clear the punch flag. Less likely, the PSF instruction skipped even though the flag was clear.
E2	0002	A spurious interrupt occurred. Less likely, the PSF instruction failed to skip when the flag was set.
E3	0117	As an error stop, this occurs when an incorrect character has been read from the tape. The erroneous code is displayed in the AC.
E3A	0123	After the preceding error, the AC displays the correct code.

Stop	C(MA)	Cause of Stop
(E3)	0117	As a legitimate halt, this signals that the program has reached the blank trailer. This is the FINAL STOP.
A	0041	Legitimate halt. This stop occurs after the trailer has been punched, to allow the operator to remove the punched tape and place it in the reader.

There are no halts in the SR Punch Test program.

3.5 Recovery from Such Errors

Error	Recovery Procedure
E1	To resume punching, press CONTINUE.
E2	If the interrupt has failed, terminate the Punch Test. Run Instruction Test 2A to determine cause.
E3	Record the C(AC). Press CONTINUE to reach the next halt.
E3A	Record the C(AC) and compare it with the previous C(AC). Check the character on tape. To resume testing, press CONTINUE.

3.5.1 Other Stops - When the program halts at A, remove the punched tape from the punch bin and place it in the reader to be used for checking. Then follow the procedure described in paragraph 3.3 to begin the checking of the tape.

4. RESTRICTIONS

If the computer has any input/output devices other than the Teletype keyboard/printer with reader and punch, the Type 750 High-Speed Reader, and the Type 75A High-Speed Punch, be sure that these devices are turned off. If there is any chance that one of them may have been left active, insert a "clear status" instruction for that device in one of the two locations reserved for this purpose: 0153 or 0154. This is in a subroutine that clears all the device flags before starting the punch program.

5. DESCRIPTION

This program tests the performance of the high-speed punch in producing a random sequence of characters. The program begins by punching a length of blank leader. It then punches a sequence of "pseudo-random" numbers (see METHODS) until the operator places SW_0 up. At this point it terminates the Punch Test, produces a length of blank trailer, and stops.

The operator removes the tape and places it in the reader. The next part of the program scans the tape until the first nonzero character appears. It then checks the "random" sequence for errors. The checking continues until an error occurs, or until the first blank character appears, signaling the end of the test section and the beginning of the trailer.

A separate section of the test program allows the operator using an oscilloscope to examine the wave forms generated by the punch when punching a single character. The start of this test is at location 0022. The program punches the character code corresponding to the setting of switches 4-11. The operator can, of course, change these settings at will.

When a single character is punched, the tape can be examined visually for proper spacing between lines as well as for accuracy.

6. METHODS

6.1 Discussion

The sequence of characters punched on the test tape is determined by a random number generating subroutine. The sequence is initialized by two random number seeds, 1233_8 and 7622_8. The reading program checks the punched sequence by comparing each character with the number generated by the same subroutine.

6.2 Algorithm(s)

The algorithm for generating the sequence of pseudo-random numbers follows.

All numbers given are octal integers. RN designates the number that is punched. RN=0 is not permitted.

$R1_0 = 1233$; $R2_0 = 7622$

$R1_{n+1} = 4(R1_n) + R2_n$ (1)

$R2_{n+1} = 4[4(R1_{n+1}) + R2_n]$ (2)

$RN_n = (R1_{n+1} \wedge 377)$ (3)

If $RN_n = 0$, then $RN_{n+1} = RN_n$ (4)

$n = 0, 1, 2, \ldots\ldots$

7. PROGRAM LISTING

```
                /PDP-8 HIGH SPEED PUNCH TEST

                *1                                  /PUNCH TEST SUBROUTINES
0001  6021              PSF
0002  7402  E2,          HLT                       /ERROR: FAILED OR SPURIOUS INTERRUPT
0003  5404              JMP I PUN

0004  0000  PUN,         Ø                         /PUNCH SUBROUTINE
0005  6026              PLS
0006  6001              ION
0007  6021              PSF                       /TEST FOR CLEARED FLAG
0010  7610              SKP CLA                   /CLEAR--GO ON
0011  7402  E1,          HLT                       /ERROR: PUNCH FLAG NOT CLEARED PLS
                                                  /OR PSF SKIPPED ON ZERO FLAG
0012  5012              JMP .                     /WAIT FOR INTERRUPT

0013  6011  CT1,         RSF                       /CONSTANTS TABLE 1
0014  6031              KSF
0015  6036              KRB

                *20
                                                  /INITIAL ENTRY POINTS
0020  5027              JMP HIGH                  /FOR HIGH SPEED READER
0021  5034              JMP LOW                   /FOR KEYBOARD READER
0022  7604              LAS                       /FOR PUNCH TEST USING SWITCHES
0023  6026              PLS
0024  6021              PSF
0025  5024              JMP .-1
0026  5022              JMP .-4

0027  1013  HIGH,       TAD CT1                   /SETUP TO USE HIGH SPEED READER
0030  3075              DCA RPT+2
0031  1102              TAD R                     /CONTAINS RFC INSTRUCTION
0032  3100              DCA R-2
0033  5041              JMP A

0034  1014  LOW,        TAD CT1+1                 /SETUP TO USE KEYBOARD READER
0035  3075              DCA RPT+2
0036  1015              TAD CT1+2
0037  3100              DCA R-2
0040  5041              JMP A

0041  7602  A,          HLT CLA                   /START OF TEST PROGRAMS
0042  4145              JMS CLRF                  /CLEAR I-O FLAGS
0043  1155              TAD CT2                   /INITIALIZE RANDOM NUMBER ROUTINE
0044  3161              DCA RAN1
0045  1155              TAD CT2
0046  3162              DCA RAN2
0047  7604              LAS
0050  7710              SPA CLA                   /TEST SW-Ø
0051  5102              JMP R                     /ON--READ PUNCHED TAPE
0052  4063              JMS FEED                  /OFF--BEGIN PUNCH TEST

0053  7604  C,          LAS
0054  7500              SMA                       /TERMINATE PUNCH TEST?
0055  5060              JMP D                     /NO--GET ANOTHER NUMBER
0056  4063              JMS FEED                  /YES--PUNCH BLANK TRAILER
0057  5041              JMP A                     /STOP
```

```
0060  4125  D,      JMS RNUM          /GET ONE NUMBER
0061  4004          JMS PUN           /PUNCH IT
0062  5053          JMP C

0063  0000  FEED,   0                 /TAPE FEED SUBROUTINE
0064  7200          CLA
0065  1157          TAD CT2+2         /COUNTER (200 LINES)
0066  3163          DCA TEM
0067  4004          JMS PUN
0070  2163          ISZ TEM
0071  5067          JMP .-2
0072  5463          JMP I FEED

0073  0000  RPT,    0                 /READ SUBROUTINES FOR CHECKING
0074  7200          CLA
0075  6011          RSF               /(OR KSF)
0076  5075          JMP .-1
0077  6012          RRB               /(NO EFFECT WITH KEYBOARD READER)
0100  6014          RFC               /(OR KRB)
0101  5473          JMP I RPT

0102  6014  R,      RFC               /ENTRY FOR READ SR
0103  4073          JMS RPT
0104  7450          SNA
0105  5103          JMP .-2           /SKIP OVER BLANK LEADER
0106  3163  SS,     DCA TEM           /STORE CHARACTER READ
0107  4125          JMS RNUM          /GET NUMBER FOR COMPARISON
0110  7041          CIA
0111  1163          TAD TEM           /(LOGICAL OR)
0112  7640          SZA CLA           /TEST RESULT OF OR
0113  5116          JMP ERR           /ERROR: AC NOT ZERO
0114  4073  S2,     JMS RPT           /OK: READ NEXT CHARACTER
0115  5106          JMP SS

0116  1163  ERR,    TAD TEM           /ERROR SR
0117  7402  E3,     HLT               /ERROR: AC CONTAINS CHAR PUNCHED
                                      /OR END OF TAPE: FINAL HALT.
0120  7200          CLA
0121  1161          TAD RAN1
0122  0160          AND CT2+3
0123  7402  E3A,    HLT               /AC CONTAINS CORRECT CHAR
0124  5114          JMP S2            /READ NEXT CHARACTER

0125  0000  RNUM,   0                 /RANDOM NUMBER GENERATOR SR
0126  7300          CLL CLA
0127  1161          TAD RAN1
0130  7006          RTL
0131  1162          TAD RAN2
0132  3161          DCA RAN1
0133  1161          TAD RAN1
0134  7006          RTL
0135  1162          TAD RAN2
0136  7006          RTL
0137  3162          DCA RAN2
0140  1161          TAD RAN1
0141  0160          AND CT2+3         /TRUNCATE NO. TO 8 BITS
0142  7450          SNA               /TEST FOR ZERO NO.
0143  7001          IAC               /IF ZERO, CHANGE TO 1
0144  5525          JMP I RNUM
```

```
0145  0000  CLRF,   0                 /CLEAR I-O FLAGS SR
0146  6012          RRB
0147  6022          PCF
0150  6042          TCF
0151  7000          NOP               /FOR USER'S SPECIAL DEVICES
0152  7000          NOP
0153  6032          KCC
0154  5545          JMP I CLRF

              CT2,                    /CONSTANTS TABLE 2
0155  1233          1233              /FIRST RANDOM NUMBER SEED
0156  7622          7622              /SECOND RN SEED
0157  7577          7577              /CHAR COUNTER FOR TAPE FEED
0160  0377          377               /RN MASK

0161  0000  RAN1,   0
0162  0000  RAN2,   0
0163  0000  TEM,    0
```

```
A       0041
C       0053
CLRF    0145
CT1     0013
CT2     0155
D       0060
ERR     0116
E1      0011
E2      0002
E3      0117
E3A     0123
FEED    0063
HIGH    0027
LOW     0034
PUN     0004
R       0102
RAN1    0161
RAN2    0162
RNUM    0125
RPT     0073
SS      0106
S2      0114
TEM     0163
```

8. FLOW CHART


START
SA:0022
SR
SA:0020
SA:0021
SETUP H.S. RDR. CHECK
SETUP TTY RDR. CHECK
HLT A
CLEAR FLAGS
INITIALIZE RAN NO.
SR0= ?
=1
=0
FEED BLANK LEADER
C
SR0= ?
=0
=1
FEED BLANK TRAILER
GENERATE RANDOM NUMBER
PUNCH IT
PUN
READ A CHAR FROM TAPE
IS CODE = 0 ?
YES
NO
GENERATE RANDOM NUMBER & COMPARE WITH TAPE CHAR
ARE THEY EQUAL ?
YES
NO
TYPE CHAR → AC
HLT E3
RANDOM NO. → AC
HLT E3A
SR
C(SR) → C(AC)
PUNCH C(AC)
PUN
PUNCH CHAR
TURN ON INTERRUPT
DID PLS CLEAR FLAG ?
NO
YES
HLT E1
DID INTERRUPT OCCUR ?
YES
NO
HLT E2
C